

Generalized Mixed Models for Logistic Clustered Outcomes using SAS PROC GLIMMIX 94 schools; 13,802 students; between 31–515 students in each school ($M = 275$)

```
* SAS: Observed frequencies for frlunch outcome;
PROC FREQ DATA=grade10; TABLE frlunch; run;
```

frlunch: 0=No, 1=Free/Reduced Lunch

frlunch	Frequency	Percent	Cumulative Frequency	Cumulative Percent
0	9059	69.25	9059	69.25
1	4023	30.75	13082	100.00

```
* Centering math predictors (had previously used PROC MEANS BY school to get school mean math);
DATA grade10; SET grade10;
  WSmath = (math - Smmath)/10;
  LABEL WSmath= "WSmath: Within-School Math (0=SM)";
  SMmath50 = (SMmath - 50)/10;
  LABEL SMmath50= "SMmath50: School Mean Math (0=5)";
RUN; PROC MEANS DATA=grade10; VAR SMmath WSmath math; RUN;
```

Variable	N	Mean	Std Dev	Minimum	Maximum
math	13082	48.1185599	17.2590473	0	83.0000000
SMmath50	13082	-0.1881440	0.6818130	-2.0549020	1.1613636
WSmath	13082	-1.31097E-17	1.5855214	-4.4650699	4.4881890

Model 1a. Empty Means Logistic Model for Paid Lunch (0) vs. Free/Reduced Lunch (1)

```
TITLE1 "SAS Empty Means, Logistic Single-Level Model for Student Free/Reduced Lunch";
PROC GLIMMIX DATA=grade10 NOCLPRINT NOITPRINT NAMELEN=100 METHOD=QUAD (QPOINTS=15) GRADIENT;
  CLASS schoolID studentID;
  * Descending makes us predict the 1 instead of the 0;
  MODEL frlunch (DESCENDING) = / SOLUTION LINK=LOGIT DIST=BINARY DDFM=Satterthwaite;
  ESTIMATE "Intercept" intercept 1 / ILINK; * ILINK is inverse link (to un-logit);
RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

Level 1: $\text{Logit}(\text{FRlunch}_{ks} = 1) = \beta_{0s}$
 Level 2: Intercept: $\beta_{0s} = \gamma_{00}$

Fit Statistics

-2 Log Likelihood	16145.89
AIC (smaller is better)	16147.89
AICC (smaller is better)	16147.89
BIC (smaller is better)	16155.37
CAIC (smaller is better)	16156.37
HQIC (smaller is better)	16150.39
Pearson Chi-Square	13082.00
Pearson Chi-Square / DF	1.00

To go from logits to probability for predicted outcomes:

$$\text{Prob}(y = 1) = \frac{\exp(-0.8117)}{1 + \exp(-0.8117)} = .3075$$

Parameter Estimates

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	-0.8117	0.01895	13081	-42.84	<.0001	2.155E-9

Label	Estimate	Standard Error	DF	t Value	Pr > t	Mean	Standard Error
Intercept	-0.8117	0.01895	13081	-42.84	<.0001	0.3075	0.004035

What does the fixed intercept represent?

Model 1b. Empty Logistic Two-Level Model for Paid Lunch (0) vs. Free/Reduced Lunch (1)

$$\text{Level 1: Logit}(FRlunch_{ks} = 1) = \beta_{0s}$$

$$\text{Level 2: Intercept: } \beta_{0s} = \gamma_{00} + U_{0s}$$

```
TITLE1 "SAS Empty Means, Logistic Two-Level Model for Student Free/Reduced Lunch";
PROC GLIMMIX DATA=grade10 NOCLPRINT NOITPRINT NAMELEN=100 METHOD=QUAD (QPOINTS=15) GRADIENT;
CLASS schoolID studentID;
MODEL frlunch (DESCENDING) = / SOLUTION LINK=LOGIT DIST=BINARY DDFM=BW;
RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
ESTIMATE "Intercept" intercept 1 / ILINK; * ILINK is inverse link (to un-logit);
COVTEST "Random School Intercept?" 0; * Test if G matrix (1,1)=0; RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

DDFM=Satterthwaite or KR is not available using METHOD=QUAD, so we have to switch to DDFM=BW (Between-Within).

Fit Statistics

-2 Log Likelihood	13172.43
AIC (smaller is better)	13176.43
AICC (smaller is better)	13176.43
BIC (smaller is better)	13181.52
CAIC (smaller is better)	13183.52
HQIC (smaller is better)	13178.48

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	schoolID	1.9545	0.3315	0.000164

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	-1.1721	0.1494	93	-7.85	<.0001	0.000085

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t	Mean	Standard Error
Intercept	-1.1721	0.1494	93	-7.85	<.0001	0.2365	0.02697

Tests of Covariance Parameters

Based on the Likelihood

Label	DF	-2 Log Like	ChiSq	Pr > ChiSq	Note
Random School Intercept?	1	16146	2973.46	<.0001	MI

MI: P-value based on a mixture of chi-squares.

The COVTEST tells us whether adding the random intercept variance across schools significantly improves model fit: -2LL single level = 16,145.89 -2LL two level = 13,172.43 -2ΔLL (df=~1) = 2,973.46

COVTEST can be used for any nested model comparisons involving variance components, but I have seen it get the answer wrong, so be careful when using it!

What does the fixed intercept NOW represent?

Note that they are also scaled differently given that there is more variance in the outcome.

To go from logits to probability for predicted outcomes:

$$\text{Prob}(y = 1) = \frac{\exp(-1.1721)}{1 + \exp(-1.1721)} = .2365$$

Calculate a 95% random effect confidence interval for the school random intercept:

$CI = \text{fixed effect} \pm 1.96 * \text{SQRT}(\text{random intercept variance})$

$CI = -1.1721 \pm 1.96 * \text{SQRT}(01.9545) = -3.91 \text{ to } 1.57 \text{ in logits, } .02 \text{ to } .83 \text{ in probability}$

Model 2a. Adding a Level-2 Fixed Effect of School Mean Student Math

$$\text{Level 1: } \text{Logit}(\text{FRLunch}_{ks} = 1) = \beta_{0s}$$

$$\text{Level 2: } \text{Intercept: } \beta_{0s} = \gamma_{00} + \gamma_{01}(\text{SMmath}_s - 50) + U_{0s}$$

```
TITLE1 "SAS Add Level-2 Fixed Effect of School Mean Math";
PROC GLIMMIX DATA=grade10 NOCLPRINT NOITPRINT NAMELEN=100 METHOD=QUAD (QPOINTS=15) GRADIENT;
  CLASS schoolID studentID;
  MODEL frlunch (DESCENDING) = SMmath50 / SOLUTION LINK=LOGIT DIST=BINARY DDFM=BW ODDSRATIO;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
  ESTIMATE "Intercept if SMmath=49" intercept 1 SMmath50 -1 / ILINK;
  ESTIMATE "Intercept if SMmath=50" intercept 1 SMmath50 0 / ILINK;
  ESTIMATE "Intercept if SMmath=51" intercept 1 SMmath50 1 / ILINK;
  ESTIMATE "L2 Math Slope" SMmath50 1 / ILINK; * Example of non-sense ILINK;
RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

Fit Statistics

-2 Log Likelihood	13103.22
AIC (smaller is better)	13109.22
AICC (smaller is better)	13109.23
BIC (smaller is better)	13116.85
CAIC (smaller is better)	13119.85
HQIC (smaller is better)	13112.31

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	schoolID	0.7657	0.1448	-0.00005

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	-1.4696	0.1040	92	-14.13	<.0001	0.000025
SMmath50	-1.4429	0.1403	92	-10.29	<.0001	-0.00002

Odds Ratio Estimates

SMmath50	_SMmath50	Estimate	DF	95% Confidence Limits	
0.8119	-0.188	0.236	92	0.179	0.312

Effects of continuous variables are assessed as one unit offsets from the mean. The AT suboption modifies the reference value and the UNIT suboption modifies the offsets.

Label	Estimates					Standard Error	
	Estimate	Standard Error	DF	t Value	Pr > t	Mean	Mean
Intercept if SMmath=49	-0.02668	0.1421	92	-0.19	0.8515	0.4933	0.03552
Intercept if SMmath=50	-1.4696	0.1040	92	-14.13	<.0001	0.1870	0.01581
Intercept if SMmath=51	-2.9125	0.2020	92	-14.42	<.0001	0.05154	0.009873
L2 Math Slope	-1.4429	0.1403	92	-10.29	<.0001	0.1911	0.02168

What does the fixed intercept NOW represent?

The logit of the probability of getting free/reduced lunch for a kid in a school with a random intercept $U_{0s} = 0$ and school mean math = 50 is -1.4696 , which translates into a probability = .187.

What does the main effect of school mean math represent?

For every 10 units higher school mean math, the logit of the probability of getting free/reduced lunch is significantly lower by 1.4429, which translates into an odds ratio of 0.236. This is NOT controlling for student math, so it is the "total" between-school effect. Note that the probability estimate of 0.1911 is meaningless, because a one-unit difference in the predictor does not imply the same difference in probability at all points along the predictor.

Model 2b. Adding a Level-1 Fixed Effect of Group-Mean-Centered Student Math

```
TITLE1 "SAS Add Level-1 Fixed Effect of Group-MC Student Math";
PROC GLIMMIX DATA=grade10 NOCLPRINT NOITPRINT NAMELEN=100 METHOD=QUAD (QPOINTS=15) GRADIENT;
  CLASS schoolID studentID;
  MODEL frlunch (DESCENDING) = SMmath50 WSmath
    / SOLUTION LINK=LOGIT DIST=BINARY DDFM=BW ODDSRATIO;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=schoolID;
  ESTIMATE "Contextual Effect of Math" WSmath -1 SMmath50 1;
RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

Fit Statistics

-2 Log Likelihood	12390.67
AIC (smaller is better)	12398.67
AICC (smaller is better)	12398.67
BIC (smaller is better)	12408.85
CAIC (smaller is better)	12412.85
HQIC (smaller is better)	12402.78

Level 1: $\text{Logit}(\text{FRlunch}_{ks} = 1) = \beta_{0s} + \beta_{1s} (\text{math}_{ks} - \text{SMmath}_s)$
 Level 2: Intercept: $\beta_{0s} = \gamma_{00} + \gamma_{01} (\text{SMmath}_s - 50) + U_{0s}$
 Within-School Math: $\beta_{1s} = \gamma_{10}$

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	schoolID	0.8414	0.1576	0.000012

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	-1.5598	0.1088	92	-14.34	<.0001	-0.00046
SMmath50	-1.5174	0.1467	92	-10.35	<.0001	0.00009
WSmath	-0.3720	0.01450	12987	-25.66	<.0001	0.000823

Odds Ratio Estimates

SMmath50	WSmath	_SMmath50	_WSmath	Estimate	DF	95% Confidence Limits	
0.8119	-1E-17	-0.188	-1E-17	0.219	92	0.164	0.293
-0.188	1	-0.188	-1E-17	0.689	12987	0.670	0.709

Effects of continuous variables are assessed as one unit offsets from the mean. The AT suboption modifies the reference value and the UNIT suboption modifies the offsets.

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Contextual Effect of Math	-1.1454	0.1468	92	-7.80	<.0001

What does the fixed intercept NOW represent?

The logit of the probability of getting free/reduced lunch for a kid in a school with a random intercept $U_{0s} = 0$ and school mean math = 50 and within-school math = 0 (e.g., an average student) is -1.5598, which translates into a probability = .210.

What does the main effect of school mean math NOW represent?

The interpretation is the same: for every one-unit higher school mean math, the logit of the probability of getting free/reduced lunch is significantly lower by 0.1517, which translates into an odds ratio of 0.219. This effect is still significant after controlling for kid math (contextual effect = -1.1454).

The between-school effect should not have changed, but the model is now on a different scale because the student math effect explained part of the level-1 residual variance (which can't be less than 3.29). This is indicated by an increase in the random intercept variance (0.8415 relative to 0.7657).

What does the main effect of student math represent?

For every 10 units higher student math relative to the rest of your school, the logit of the probability of getting free/reduced lunch is significantly lower by 0.372, which translates into an odds ratio of 0.689.

Model 2c. Adding a Random Effect of Group-MC Student Math

```
TITLE1 "SAS Add Random Effect of Group-MC Student Math";
PROC GLIMMIX DATA=grade10 NOCLPRINT NOITPRINT NAMELEN=100 METHOD=QUAD (QPOINTS=15) GRADIENT;
  CLASS schoolID studentID;
  MODEL frlunch (DESCENDING) = SMmath50 WSmath
    / SOLUTION LINK=LOGIT DIST=BINARY DDFM=BW ODDSRATIO;
  RANDOM INTERCEPT WSmath / TYPE=UN SUBJECT=schoolID;
  COVTEST "Random Student Math Slope?" . 0 0; * Leave (1,1), test if (2,1) and (2,2) =0;
RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

FROM THE LOG: At least one element of the gradient is greater than 1e-3.

Fit Statistics

-2 Log Likelihood	12352.01
AIC (smaller is better)	12364.01
AICC (smaller is better)	12364.01
BIC (smaller is better)	12379.27
CAIC (smaller is better)	12385.27
HQIC (smaller is better)	12370.17

Level 1: $\text{Logit}(\text{FRLunch}_{ks} = 1) = \beta_{0s} + \beta_{1s} (\text{math}_{ks} - \text{SMmath}_s)$

Level 2: Intercept: $\beta_{0s} = \gamma_{00} + \gamma_{01} (\text{SMmath}_s - 50) + U_{0s}$

Within-School Math: $\beta_{1s} = \gamma_{10} + U_{1s}$

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	schoolID	0.8118	0.1540	-0.00188
UN(2,1)	schoolID	-0.03524	0.02906	0.007376
UN(2,2)	schoolID	0.01608	0.005433	0.324555

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	-1.5665	0.1076	92	-14.56	<.0001	0.003945
SMmath50	-1.5617	0.1477	92	-10.57	<.0001	-0.0015
WSmath	-0.3434	0.02425	12987	-14.16	<.0001	-0.04844

Odds Ratio Estimates

SMmath50	WSmath	_SMmath50	_WSmath	Estimate	DF	95% Confidence Limits	
0.8119	-1E-17	-0.188	-1E-17	0.210	92	0.156	0.281
-0.188	1	-0.188	-1E-17	0.709	12987	0.676	0.744

Effects of continuous variables are assessed as one unit offsets from the mean. The AT suboption modifies the reference value and the UNIT suboption modifies the offsets.

Tests of Covariance Parameters

Label	DF	-2 Log Like	ChiSq	Pr > ChiSq	Note
Random Student Math Slope?	2	12391	38.66	<.0001	MI

MI: P-value based on a mixture of chi-squares.

Does the random effect of student math improve model fit?

Yes, $-2\Delta LL(\sim 2) = 38.66, p < .00$ (but it is not estimated very well)

Calculate a 95% random effect confidence interval for the student math slope:

$CI = \text{fixed effect} \pm 1.96 * \text{SQRT}(\text{random slope variance})$

$CI = -0.3434 \pm 1.96 * \text{SQRT}(0.01608) = -0.59 \text{ to } -0.09 \text{ in logits}$

So what does this mean?

The extent to which relative student math predicts student free/reduced lunch status varies significantly across schools, but across 95% of schools, higher student math is predicted to relate to a lower probability of receiving free/reduced lunch.

Model 2d. Adding Intra-Variable Interactions of School Mean Math and GMC Student Math

```
TITLE1 "SAS Add Intra-Variable Interactions of School Mean and Group-MC Student Math";
PROC GLIMMIX DATA=grade10 NOCLPRINT NOITPRINT NAMELEN=100 METHOD=QUAD (QPOINTS=15) GRADIENT;
CLASS schoolID studentID;
MODEL frlunch (DESCENDING) = SMmath50 WSmath SMmath50*WSmath SMmath50*SMmath50
/ SOLUTION LINK=LOGIT DIST=BINARY DDFM=BW ODDSRATIO;
RANDOM INTERCEPT WSmath / TYPE=UN SUBJECT=schoolID;
ESTIMATE "Contextual Math Main Effect" WSmath -1 SMmath50 1;
ESTIMATE "Contextual Math Interaction" SMmath50*WSmath -1 SMmath50*SMmath50 1;
RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.
FROM THE LOG: At least one element of the gradient is greater than 1e-3.

Fit Statistics

-2 Log Likelihood	12347.84
AIC (smaller is better)	12363.84
AICC (smaller is better)	12363.86
BIC (smaller is better)	12384.19
CAIC (smaller is better)	12392.19
HQIC (smaller is better)	12372.06

Level 1: $\text{Logit}(\text{FRlunch}_{ks} = 1) = \beta_{0s} + \beta_{1s} (\text{math}_{ks} - \text{SMmath}_s)$

Level 2: Intercept: $\beta_{0s} = \gamma_{00} + \gamma_{01} (\text{SMmath}_s - 50) + \gamma_{02} (\text{SMmath}_s - 50)^2 + U_{0s}$

Within-School Math: $\beta_{1s} = \gamma_{10} + \gamma_{11} (\text{SMmath}_s - 50) + U_{1s}$

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	schoolID	0.8157	0.1553	-0.00526
UN(2,1)	schoolID	-0.02773	0.02798	-0.05393
UN(2,2)	schoolID	0.01348	0.004909	0.332867

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	-1.5460	0.1231	91	-12.55	<.0001	0.003075
SMmath50	-1.5833	0.1998	91	-7.93	<.0001	-0.00215
WSmath	-0.3688	0.02633	12986	-14.01	<.0001	-0.10677
SMmath50*WSmath	-0.06962	0.03364	12986	-2.07	0.0385	0.055708
SMmath50*SMmath50	-0.06850	0.1760	91	-0.39	0.6980	0.0059

Odds Ratio Estimates

SMmath50	WSmath	_SMmath50	_WSmath	Estimate	DF	95% Confidence Limits	
0.8119	-1E-17	-0.188	-1E-17	0.197	91	0.111	0.348
-0.188	1	-0.188	-1E-17	0.701	12986	0.668	0.735

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Contextual Math Main Effect	-1.2145	0.1994	91	-6.09	<.0001
Contextual Math Interaction	0.001114	0.1772	91	0.01	0.9950

What does from Within-School*Between-School math interaction represent?
For every 10 units higher school mean math, the effect of relative student math on student free/reduced lunch (which is -0.3688 as evaluated at school mean math = 50) becomes significantly more negative by 0.06962. So the effect of being "smarter than the others" is even stronger in a "smart" school.



What does from Between-School*Between-School math interaction represent?
*For every 10 units higher school mean math, the effect of school mean math on school mean free/reduced lunch (which is -1.5833 as evaluated at school mean math = 50) becomes nonsignificantly more negative by 2*0.06850. So the effect of being in a "smart" school is predominantly linear. These effects do not control for student math.*

What do the contextual math effects represent?

After controlling for student math, there is a contextual effect of school mean math 1.2145 per 10 units as evaluated at school mean math = 50 for an average student. However, there is not a contextual effect of how school mean math moderates the effect of relative student math (incremental interaction = 0.0011).

—OR—

The between-school math effect is significantly more negative by 1.2145 as evaluated at school mean math = 50 for an average student. However, school mean math does not moderate the between-school math effect (-0.06850) differently than the within-school math effect (-0.06962).



Mplus Syntax and Output for final MLM (using observed variables as predictors rather than latent) – results are very similar to SAS:

<pre> TITLE: 2-Level Model for Students within Schools Predicting FR Lunch; DATA: FILE = graded10M.csv; ! Can just list file if in same directory; FORMAT = free; ! FREE or FIXED format; TYPE = individual; ! Individual or matrix data as input; VARIABLE: ! List of ALL variables in stacked data file, in order; ! Mplus does NOT know what they used to be called, though; NAMES ARE Student School BvG FRLunch Math smvG smFR smMath SchoolN smBvG50 smFR30 WSmath smMath50; ! List of ALL variables used in model (DEFINED variables at end); USEVARIABLES ARE FRLunch WSmath smMath50 smMath2; ! Outcomes that are binary/ordinal; CATEGORICAL ARE FRLunch; ! Missing data codes (here, -999); MISSING ARE ALL (-999); ! Identify person-level nesting; CLUSTER = School; ! Predictor variables with variation ONLY within-persons at level 1; WITHIN = WSmath; ! Predictor variables with variation ONLY between-persons at level 2; BETWEEN = smMath50 smMath2; DEFINE: smMath2 = smMath50*smMath50; ! Creating level-2 math quadratic; ANALYSIS: TYPE IS TWOLEVEL RANDOM; ! 2-level model with random slopes; ESTIMATOR IS ML; ! Can also use MLR for non-normality; MODEL: !!! MODEL 2d ! Level-1, child-level model; %WITHIN% ! NO residual variance is estimated for FRLunch at level 1; Llmath FRLunch ON WSmath; ! Bls effect of 0/1 level-1 FRLunch; ! Level-2, school-level model; %BETWEEN% FRLunch; ! Random intercept variance (is default); [FRLunch\$1]; ! Fixed "threshold" (is intercept*-1); [Llmath] (L1math); ! Fixed WS effect of level-1 math; L1math; ! Yes random effect of level-1 math; FRLunch WITH L1math; ! Covariance of intercept and math slope; FRLunch ON smMath50 (L2math); ! Linear BS math on intercept; FRLunch ON smMath2 (L2math2); ! Quad BS math on intercept; L1math ON smMath50 (L12math); ! Cross-level L1 by L2 math interaction; !!!! Adding NEW statements to show how to get ESTIMATE-type statements; MODEL CONSTRAINT: ! Define new parameters not directly given by model; NEW (conM conMint); conM = L2math - L1math; ! Contextual main effect of math; conMint = L2math2 - L12math; ! Contextual L2 interaction of math; </pre>	<pre> UNIVARIATE PROPORTIONS AND COUNTS FOR CATEGORICAL VARIABLES FRLUNCH Category 1 0.692 9059.000 Category 2 0.308 4023.000 THE MODEL ESTIMATION TERMINATED NORMALLY MODEL FIT INFORMATION Number of Free Parameters 8 Loglikelihood H0 Value -6173.936 Information Criteria Akaike (AIC) 12363.871 Bayesian (BIC) 12423.703 Sample-Size Adjusted BIC 12398.280 (n* = (n + 2) / 24) MODEL RESULTS Estimate S.E. Est./S.E. Two-Tailed P-Value Within Level Between Level L1MATH ON SMMATH50 -0.069 0.034 -2.065 0.039 FRLUNCH ON SMMATH50 -1.587 0.200 -7.952 0.000 SMMATH2 -0.083 0.176 -0.472 0.637 FRLUNCH WITH LIMATH -0.027 0.028 -0.972 0.331 Intercepts LIMATH -0.369 0.026 -14.099 0.000 Thresholds FRLUNCH\$1 1.526 0.123 12.443 0.000 Residual Variances FRLUNCH 0.813 0.155 5.251 0.000 LIMATH 0.013 0.005 2.729 0.006 New/Additional Parameters CONM -1.218 0.199 -6.115 0.000 CONMINT -0.014 0.177 -0.077 0.939 </pre>
--	--